

finish \$/clear

!-----PARAMETERFESTLEGUNG-----

!-----Geometrie-----

berad=12 !Bereichs-Radius

!-----Schaltungsparameter-----

kapazit=500e-6 !Kapazität

spannung=-10e3 !Ladespannung

widerst=2e-3 !Vorwiderstand

kappa=56e6 !spez. el. Leitfähigkeit des Leiters

mw=2 !Maschenweite

abm=0.075 !Schaltungsabmessungen

!/eof

!-----MODELLGENERIERUNG-----

!-----Elementetypen-----

/prep7 !Aufruf des Preprozessors

et,1,solid97 !Luft

et,2,solid97,4 !Leiter

et,3,infin111,1 !Randelement

et,4,circu124,2 !Kondensator

et,5,circu124,0 !Widerstand

et,6,circu124,7 !Koppelement

!/eof

!-----Real-Konstanten-----

r,2,1*5 !Querschnitt

r,4,kapazit,spannung

r,5,widerst

r,6,1

!-----Materialeigenschaften-----

mp,murx,1,1

!rel. Permeabilität (Luft)

mp,murx,2,1

!rel. Permeabilität (Spule)

mp,rsvx,2,1/kappa

!spez. el. Leitfähigkeit (Spule)

!-----

/view,,1,2,3

/plopts,info,1

!-----Geometrie-----

cyl4,0,5,2.5,0,3.5,90,1

cyl4,0,5,2.5,90,3.5,180,1

BLC4,2.5,0,1,5,1
BLC4,-3.5,0,1,5,1
vglue,1,2,3,4

sphere,berad
block,-3,3,3,9,3,2
vovlap,all

!-----Vernetzung des Leiters-----

esize,mw
mshkey,1 !mapped
mshape,0,3D
type,2 \$mat,2 \$real,2
vmesh,1

!-----Vernetzung-Luft-----

csys,1
asel,s,loc,x,berad
aesize,all,mw*6
allsel
type,1 \$mat,1
mshkey,0
mshape,1,3D !Tetraeder
vmesh,3

!-----Randlelemente-----

csys,1
type,3
esize,,1
asel,s,loc,x,berad
*get,anr1,area,,num,min !A9
*get,anr2,area,,num,max !A10
vext,anr1,anr2,,mw
nsel,s,loc,x,berad+mw/2,berad+mw
sf,all,inf
csys,0
allsel

!-----Randbedingungen-- -----

!-----vordere Leiterquerschnittsfläche-----

esel,s,mat,,2
nsle
nsel,r,loc,z,2.5
cp,1,curr,all
cp,2,volt,all
sf,all,mci,-1 !Stromfluß in diese Stirnfläche hinein (-1)
*get,knr1,node,0,num,max
!/eof

!-----hintere Leiterquerschnittsfläche-----

esel,s,mat,,2

nsle

nsel,r,loc,z,-3.5

cp,3,curr,all

cp,4,volt,all

sf,all,mci,+1 !Stromfluß aus dieser Stirnfläche heraus (+1)

!d,all,volt,0

*get, knr2,node,0,num,max

!-----Symmetrie-----

nsel,s,loc,z,-3.5

dsym,asymm,z

nsel,s,loc,z,2.5

dsym,asymm,z

!-----Speisende Schaltung-----

csys,0

allsel

*get, knr,node,,num,max

n, knr+1,-2*abm,0

n, knr+2,-2*abm,abm

n, knr+3,-abm,abm

n, knr+4,-abm,0

type,4 \$real,4 \$e, knr+1, knr+2 !Kondensator

type,5 \$real,5 \$e, knr+2, knr+3 !Widerstand

type,6 \$real,6 \$e, knr+3, knr+4, knr1, knr2

eplot

!/eof

!-----Bezugspotential-----

nsel,s,node,, knr+1

nsel,a,node,, knr+4

d,all,volt,0

allsel

!-----LÖSUNG-----

/solu

!Aufruf des Lösungsprozessors

allsel

!alle Selektierungen aufheben

antype,trans

!transiente Lösungsberechnung

outres,all,all

!Abspeicherung aller Ergebnisse

kbc,1 ! (0) rampenförmige (1) sprungförmige Lastschrittveränderung

endzeit=5e-6

!Berechnungszeit

schritte=10

!Berechnungsschritte

*do,i,0,schritte

time,endzeit*i/schritte+1e-12

solve

*enddo